

Ingenieurbüro Baumann --- www.leobaumann.de --- Markt 6, 46282 Dorsten

manuelle Berechnung eines 4-Lambda-Adcock Antennenpaars

h = Länge, b_2 = Höhe über Grund, λ = Wellenlänge


- `reset () :digits:=16:wv:=90*PI/180:wh:=90*PI/180:h:=1/2:d:=4:b2:=0:l:=1:`

Richtdiagramm im Kugelraum als Funktion der Winkel

- `c:=(the,phi1) -> abs((cos(PI*h/l*cos(phi1))-cos(PI*h/l))/sin(phi1)) *2*abs(cos(PI*d/l*cos(the)*sin(phi1))) *2*abs(cos(PI*2*(b2+h/2)/l*cos(phi1))):`


Horizontaldiagramm

- `plot(plot::Polar([c(the,wv),the], the = 0..2*PI, TicksNumber=None, Scaling=Constrained, AdaptiveMesh=4));`



horizontale relative Strahlungsleistungsdichte

- `plotfunc2d(c(the*PI/180,wv)^2, the = 0..180):`



Maximalwert der relativen Stahlungsleistungsdichte , auch in dBi

- ```
ghmax:=0:ghwmax:=0:for m from 0 to 2880 step 1 do
gh:=float(c(m*PI/5760,wv)^2);
if gh>ghmax then
 ghmax:=gh;
 ghwmax:=float(m/32);
end_if;
end_for:ghmax;float(10*ln(ghmax)/ln(10)+2.15);ghwmax;
```


16.0

14.19119983

0.0


Vertikaldiagramm

- ```
plot(plot::Polar([c(wh,phi1),phi1+PI/2], phi1 = -PI/2..PI/2,
TicksNumber=None, Scaling=Constrained, AdaptiveMesh=4));
```



vertikale relative Strahlungsleistungsdichte

- `plotfunc2d(c(wh,phi1*PI/180)^2, phi1 = 0..180):`



Maximalwert der relativen Stahlungsleistungsdichte , auch in dBi


- `gvmax:=0:gvwmax:=0:for m from 1 to 2880 step 1 do
gv:=float(c(wh,m*PI/5760)^2);
if gv>gvmax then
 gvmax:=gv;
 gvwmax:=float(m/32);
end_if;
end_for:gvmax;float(10*ln(gvmax)/ln(10)+2.15);gvwmax;`

16.0

14.19119983

90.0

- ```
graph:=plot::Surface([cos(the)*sin(phi1)*c(the,phi1),sin(the)*sin(phi1)*c(the,phi1),cos(phi1)*c(the,phi1)],the=0..2*PI, phi1=-PI/2..PI/2,Axes=Origin, TicksNumber=None, Scaling=Constrained, AdaptiveMesh=4):
```
- ```
plot(graph);
```



•